
フローコントロールを使っての分枝  

フローコントロールを使っての分枝: イントロダクション 

多くフィーチャーのあるパーツをお持ちで、しかし、そのうちの幾つかのフィーチャーを何度も繰

り返して測定し、これらのフィーチャーの総合的な一連の統計データを入手したい、とします。ユ

ーザーからの対応次第で、測定ルーチン内の特定の部分にジャンプしたいとします。フローコント

ロール 
コマンドを用いて、これらとその他の多くの作業を行うことができます。特定のコマンドの条件の

セットアップによって、測定ルーチンのフローをコントロールすることが可能です。 

本章はそのようなタスクを実行するために必要な情報を提供します。ここでは構文の条件文、ルー

プ、サブルーチンを説明します。また、多くのコードサンプルも提供します。 

注記: 
コードのサンプルにおいて、ループや分枝がある場合、特定の条件に割り当てられたステートメン

トをはっきりと表示するために、インデントが使用されています。実際の編集ウィンドウでは、こ

のようなインデントは見られません。 

この章では、主に以下のトピックについて説明します。 

• コントロール ペアの使用 

• 汎用ループの使用 

• ラベルの作成 

• エラー時の分岐 

• サブルーチンを用いた分枝 

• 測定ルーチンの終了 

コントロール ペアの使用 

挿入 | フロー制御コマンド | コントロール 
ペアサブメニューには、様々なペアのコマンドがあり、それらは編集ウィンドウ内で、測定ルーチ

ンの適切なフローを制御、または、「コントロール」します。編集ウィンドウ内にコントロール 
ペア 
タイプのコマンドを挿入するには、単にコマンドを入力するか、または、このサブメニューからコ

マンドを選択して下さい。 

重要: 「はい / 
いいえ」コメントの値をテストするために条件分枝ステートメントを使用する場合、そのテストは

大文字の「YES」、または、「NO」を検索します。小文字の「yes」、または、「no」では機能し

ません。コメントに関する情報については、「レポート 
コマンドの挿入」の章の「プログラマのコメントを挿入」トピックを参照して下さい。 

  

If / End If 

挿入 | フローコントロール コマンド | コントロール ペア |If / End If メニュー 
オプションを選ぶと、測定ルーチンに、条件ブロックを付け加えることができます。IFコマンドと

END 
IFコマンドの間の項目は、IFコマンドの式が真（ゼロ以外）と出た場合にのみ、実行されます。そ

れ以外の場合、実行フローは、 END/IF コマンドの後、最初のコマンドへジャンプします。 

IF / END 
IFステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります: 
 

IF/式 
END_IF/ 

「IF/ END IF」コマンドを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。 

2. メニュー バーからIf / End If を選択して下さい。IF / END IF 
ステートメントが編集ウィンドウに表示されます。 

  

  

If / End If のコードのサンプル 

点フィーチャーを測定するか否か、ユーザーに尋ねる、以下の例を参考にして下さい。 

C1= COMMENT/YESNO,点フィーチャーPNT1を測定しますか? 

IF/C1.INPUT=="YES" 

PNT1=FEAT/POINT,RECT  

… 

… 

ENDMEAS/ 

END_IF/ 

サンプル コードの説明 
C1=COMMENT/YESNO 
この行は、ユーザーからの、「はい」、または、「いいえ」の応答を受けて格納します。 
IF/C1.INPUT=="YES" 
この行は式です。これは、コメント1の入力が、「はい」であるかどうかを検査します。「はい」

である場合、IFステートメントは「真」であり、IFステートメント後のステートメントを実行し続

け、この場合、PNT1 

要素が測定されます。「いいえ」の場合、END_IFステートメントに移動します。 
END_IF 
この行は、「IF / END IF」コード 
ブロック内のコマンドの実行を終了します。ユーザーが、コメントについていいえをクリックする

と、PC-DMISは、この行に続くコマンドにジャンプします。 
Else If / End Else If 

挿入 | フローコントロール コマンド | コントロール ペア | Else If / End Else If メニュー 
オプションを選ぶと、測定ルーチンに、条件ブロックを付け加えることができます。ELSE 
IFコマンドとEND ELSE IFコマンドの間の項目は、ELSE 
IFコマンドの式が真（ゼロ以外）と出た場合にのみ、実行されます。「ELSE IF / END ELSE 
IF」ブロックは、「IF / END IF」または別の「ELSE IF / END ELSE 
IF」ブロックの直後に位置しなければなりません。その時点でのブロックより上の「IF / END 
IF」式すべてが偽と出た場合、その式が評価されます。式の結果が偽(ゼロ)の場合、END ELSE 
IFコマンドに続く次のコマンドにジャンプして実行します。現在のブロックより前のIF / ELSE if 
式のうち1つでも真と出た場合、この後に続くすべてのELSE IF / END ELSE 
IFブロックはスキップされます。 

ELSE IF / END ELSE 
IFステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります:  
 

ELSE_IF/式 
END_ELSE_IF/ 

ELSE IF / END ELSE IFコマンドを挿入するには: 

1. 編集ウィンドウ内で、既存の IF/END IFステートメント、または、ELSE IF/END ELSE IF 
ステートメントの後の、ご希望の位置にカーソルを配置して下さい。 

2. メニュー バーからElse If / End Else If を選択して下さい。ELSE IF / END ELSE IF 
ステートメントが編集ウィンドウに表示されます。 



フローコントロールを使っての分枝 

4 

注記: このタイプのブロックは、「IF / END IF」、または、「ELSE IF / END ELSE 
IF」ブロックの後に位置付けされた時のみ有効です。無効な位置付けがなされているコントロール 
ペアは、編集ウィンドウ内で赤色のテキストで表示されます。 

  

  

Else If / End Else If のコードのサンプル 

測定された点のX、Y、または、Z値のうちのいずれかが、定義された公差を越える場合、ユーザー

にメッセージを表示する、以下の例を参考にして下さい。 

PNT2=FEAT/POINT,RECT 

… 

… 

測定の終了/ 

IF/PNT2.X<6.9 OR PNT2.X>7.1 

COMMENT/OPER,"PNT2のXの測定値: " + PNT2.X + "が公差範囲外です。" 

END_IF/ 

ELSE_IF/PNT2.Y<3.3 OR PNT2.Y>3.5 

COMMENT/OPER,"PNT2のYの測定値: " + PNT2.Y + "が公差範囲外です。" 

END_ELSEIF/ 

ELSE_IF/PNT2.Z<.9 OR PNT2.Z>1.1 

COMMENT/OPER,"PNT2のZの測定値: " + PNT2.Z + "が公差範囲外です。" 

END_ELSEIF/ 

サンプル コードの説明 

このコードは、最初のポイントのX値をテストします。条件評価が 
falseだったら、コードはY値にテストします。Y 値の条件評価が falseだったら、それはZ 
値にテストします。 

これらの条件のうち、どれかが真と評価された場合、PC-
DMISは、それに関連したコメントを表示し、残りの条件文を省略します。 

IF/PNT2X7.1 
この行は式です。これは、測定されたX値が6.9より小さいか7.1より大きいかを検査します。これ

らの境界の内のどちらかを越える場合には、最初のコメントが実行されます。 
END_IF 
この行は、「IF / END IF」コード ブロック内のコマンドの実行を終了します。IF THEN 
条件が、偽と評価された場合、PC-DMISは、このラインに続くコマンドにジャンプします。 
ELSE_IF/PNT2.Y3.5 
この行は、最初の ELSE_IF 
コマンドの式です。これは、測定されたY値が3.3より小さいか、または、3.5より大きいかを検査

します。これは、測定されたY値が3.3より小さいか、または、3.5より大きいかを検査します。こ

れらの境界の内のどちらかを越える場合には、2番目のコメントが実行されます。 
END_ELSEIF/ 
この行ラインは、最初の「ELSE IF / END ELSE IF」コード 
ブロック内のコマンドの実行を終了します。 
ELSE_IF/PNT2.Z1.1 
この行は、2番目のELSE IF コマンドの式です。上の ELSE IF / END ELSE IF 
ブロックが偽と出た場合にのみ、これが実行されます。測定されたZの値が.9より小さいかまたは1
.1より大きいかを検査します。これらの境界の内のどちらかを越える場合には、3番目のコメント

が実行されます。 
END_ELSEIF/ 
この行は、2番目の「ELSE IF / END ELSE IF」コード ブロック内のコマンドの実行を終えます。 
  

Else / End Else 

挿入 | フローコントロール コマンド | コントロール ペア | Else / End Elseメニュー 
オプションを選ぶと、測定ルーチンに、条件ブロックを付け加えることができます。ELSE 
ブロックの上方にある、その他の「IF / END IF」ブロックと「ELSE IF / END ELSE 
IF」ブロックすべてが失敗の場合（すべてがゼロと出る）にのみ、ELSEコマンドとEND 
ELSEコマンドの間の項目は実行されます。「ELSE / END 
ELSE」ブロックが有効になるためには、一連の「IF / END IF」または「ELSE IF / END ELSE 
IF」ブロックの末尾に位置しなければなりません。 

ELSE / END 
ELSEステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります: 
 
ELSE/ 
END_ELSE/ 

「ELSE/ END_ELSE/」コマンドを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。「ELSE / END 
ELSE」ブロックは、「IF / END IF」または「ELSE IF / END ELSE 
IF」ブロックの後に位置つけられなければならないことに注意して下さい。 

2. メニュー バーからElse / End Else を選択して下さい。ELSE / END ELSE 
ステートメントが編集ウィンドウに表示されます。 

  

  

Else / End Else のコードのサンプル 

点フィーチャーを測定するか否か、ユーザーに尋ねる、以下の例を参考にして下さい。 

C1= COMMENT/YESNO,点フィーチャーPNT1を測定しますか? 

「いいえ」をクリックすると、次のフィーチャーが測定されます。 

IF/C1.INPUT=="YES" 

PNT1=FEAT/POINT,RECT  

… 

… 

ENDMEAS/ 

END_IF/ 

ELSE 

PNT2=FEAT/POINT,RECT 

… 

… 

測定の終了/ 

END_ELSE 

サンプル コードの説明 
C1=COMMENT/YESNO 
この行は、ユーザーからの、「はい」、または、「いいえ」の応答を受けて格納します。 
IF/C1.INPUT=="YES" 
この行は式です。これは、コメント1の入力が、「はい」であるかどうかを検査します。「はい」

である場合、IFステートメントは「真」であり、IFステートメント後のステートメントを実行し続

け、この場合、PNT1 

要素が測定されます。「いいえ」の場合、END_IFステートメントに移動します。 
END_IF 
この行は、「IF / END IF」コード 
ブロック内のコマンドの実行を終了します。ユーザーが、コメントについていいえをクリックする

と、PC-DMISは、この行に続くコマンドにジャンプします。 
ELSE 
上記の「IF / END 
IF」ブロックが偽と評価すると、このラインの後にある、「END_ELSE」ラインの前にあるコマン

ドラインが実行されます。この例の場合、PNT2が実行されます。 
END_ELSE 
この行は、2番目の「ELSE / END_ELSE」コード ブロック内のコマンドの実行を終えます。 
While / End While 

挿入 | フローコントロール コマンド | コントロール ペア |While / End Whileメニュー 
オプションを選ぶと、測定ルーチンに条件ループを追加することができます。WHILEコマンドとE
ND 
WHILEコマンドの間の項目は、ループをアクティブに保つ条件（または、式）が満たされなくなる

まで、つまり、WHILEループの式が偽（言い換えれば、ゼロ）の評価を出すまで、ループ内で実行

され続けます。WHILEコマンドは、測定ルーチン内のどこにでも追加できます。式は、各ループの

冒頭で検査されます。 

WHILE / END 
WHILEステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります

: 
 

WHILE/式 
END_WHILE/ 

「WHILE / END WHILE」オプションを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。 

2. メニュー バーからWhile / End While を選択して下さい。WHILE / END WHILE 
ステートメントが編集ウィンドウに表示されます。 

  

  

While / End While のコードのサンプル 

測定ルーチンのユーザーによって指定された分量に限定して、要素を測定する、以下の例を参考に

して下さい。 

C1=COMMENT/INPUT,PNT1を何度測定したいですか? 整数のみをタイプ入力して下さい。 

ASSIGN/COUNT = 0 

WHILE/COUNT < C1.INPUT 

PNT2=FEAT/POINT,RECT 

… 

… 

… 

ENDMEAS/ 

ASSIGN/COUNT = COUNT + 1 

COMMENT/OPER,"測定値 " + COUNT + " のうち " + C1.INPUT + " 回"。 

END_WHILE/ 

サンプル コードの説明 
C1=COMMENT/INPUT 

この行は、ユーザーから整数の入力値を受け取り、変数C1.INPUT内に保持します。 

ASSIGN/COUNT = 0 
この行はユーザー定義変数のCOUNTを初期化し、初期値0を与えます。コードはこの変数を

使用してPC-DMISがループ内で要素を測定した回数を数えます。 

WHILE/COUNT < C1.INPUT 
この行は式です。 
COUNT値（初期設定0）が、ユーザーによって選択された整数以下であるか、検査します。

これが真と出るとWHILE/の後にありEND_WHILE/の前にあるステートメントが実行されま

す。 

ASSIGN/COUNT = COUNT + 1 
この行はCOUNT変数をひとつずつ増分し、それによって、それが条件テストに失敗した時、

ループの外に出ます。 

COMMENT/OPER,"測定値 " + COUNT + " のうち " + C1.INPUT + " 回"。 
この行は、合計回数のうち、ループが実行中の回数を示すメッセージを表示します。 

END_WHILE 
このラインは、条件が偽である限り、「WHILE / END 

WHILE」ブロック内のコマンドの実行を終了します。それ以外の場合には、PC-
DMISがこのコマンドに出会うと、ループがWHILE ステートメントに戻ります。 

Do / Until 

挿入 | フローコントロール コマンド | コントロール ペア | Do / Until メニュー 
オプションを選ぶと、測定ルーチンに、条件ループを追加することができます。DOコマンドとUN
TILコマンドの間の項目は、UNTILコマンドの式が真（ゼロ以外）と出るまで、ループ内で実行さ

れ続けます。DO / 
UNTILコマンドは、測定ルーチン内のどこにでも追加できます。式は、各ループの最後で検査され

ます。 

DO / 
UNTILステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります: 
 
DO/  

UNTIL/ 式 

DO / UNTILコマンドを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。 

2. メニューバーからDo / Until を選択します。DO / UNTIL 
ステートメントが編集ウィンドウ内に表示されます。 

  

  

Do / Until のコードのサンプル 

測定ルーチンのユーザーによって指定された分量に限定して、要素を測定する、以下の例を参考に

して下さい。これは、While / End While の項目での例と類似していますが、PC-
DMISが、ループの冒頭のかわりに、その末尾で条件をテストする点が異なります。 

C1=    COMMENT/INPUT, PC-

DMISがPNT1フィーチャーを測定すべき回数をタイプ入力して下さい:(整数のみ入力) 

ASSIGN/COUNT = 0 

DO/ 

PNT1=FEAT/POINT,RECT 

… 

… 

測定の終了/ 

ASSIGN/COUNT = COUNT + 1 

COMMENT/OPER,"測定値 " + COUNT + " のうち " + C1.INPUT + " 回"。 

UNTIL/COUNT == C1.INPUT 

サンプル コードの説明 
C1=COMMENT/INPUT 
この行は、ユーザーから整数の入力値を受け取り、変数C1.INPUT内に保持します。 

ASSIGN/COUNT = 0 
この行はユーザー定義変数のCOUNTを初期化し、初期値0を与えます。コードはこの変数を

使用してPC-DMISがループ内で要素を測定した回数を数えます。 

DO/ 
DO / UNTIL 
ループを開始します。すべてのステートメントは、少なくとも一度は実行され、式が偽と評

価した場合、測定ルーチン フローはループの外に出ます。 

ASSIGN/COUNT = COUNT + 1 
この行はCOUNT変数をひとつずつ増分し、それによって、それが条件テストに失敗した時、

ループの外に出ます。 

COMMENT/OPER,"測定値 " + COUNT + " のうち " + C1.INPUT + " 回"。 
この行は、合計回数のうち、ループが実行中の回数を示すメッセージを表示します。 

UNTIL/COUNT == C1.INPUT 
この行は、条件が偽と出た場合に、DO / UNTIL 
ループ内でのコマンド実行を終了します。それ以外の場合には、PC-
DMISがこのコマンドに出会うと、ループはDO ステートメントに戻ります。 

Select / End Select 

挿入 | フロー制御コマンド | コントロール ペア| Select / End Select メニュー 
オプションを選ぶと、 CASE / END CASEとDefault Case / End Default Case 
ペアとの組み合わせで使用される条件ブロックの追加ができます。Selectコマンドの式はそのCase
ステートメントの式と比較されているデータを提供します。この二つの式が同じものと評価した場

合、 Case / End Case Blockのステートメントが実行されます。「DEFAULT CASE /END 
DEFAULT CASE」ブロックはCASE / END CASEとDEFAULT CASE / END DEFAULT CASE 
ブロックのセットを囲みます。 

SELECT / END 
SELECTステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになりま

す: 

SELECT/式 
END_SELECT/ 

「SELECT/ END SELECT」コマンドを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。 

2. メニューバーからSelect / End Select を選択します。SELECT / END SELECT 
ステートメントが編集ウィンドウ内に表示されます。 

  

  

Select / End Select のコードのサンプル 

以下のペア SELECT / END_SELECT、CASE / END_CASE、DEFAULT CASE / END_DEFAULT 

CASEはすべて一緒に機能し、代替手段を持つ、複数の条件を評価します。 

CIR1からCIR5までのラベルの付いた、5つの円があり、キーボード上のキー入力によって、簡単に

円が測定できるオペレーターが望まれる、とします。以下のようなコードを使用することができま

す: 

コード全体 
DO/ 

C1=COMMENT/INPUT,測定する円の番号をタイプ入力して下さい: 

、CIR1には - 1をタイプ入力 

、CIR2には - 2をタイプ入力 

、CIR3には - 3をタイプ入力 

、CIR4には - 4をタイプ入力 

、CIR5には - 5をタイプ入力 

それ以外の活字の場合、ループを退く 

SELECT/C1.INPUT 

CASE/1 

CIR1=FEAT/CIRCLE 

… 

… 

測定の終了/ 

END_CASE 

CASE/2 

CIR2=FEAT/CIRCLE 

… 

… 

測定の終了/ 

END_CASE 

CASE/3 

CIR3=FEAT/CIRCLE 

… 

… 

測定の終了/ 

END_CASE 

CASE/4 

CIR4=FEAT/CIRCLE 

… 

… 

測定の終了/ 

END_CASE 

CASE/5 

CIR5=FEAT/CIRCLE 

… 

… 

測定の終了/ 

END_CASE 

DEFAULT CASE 

COMMENT/OPER,ここでループを退く。 

END_DEFAULT CASE 

END_SELECT 

UNTIL C1.INPUT < 1 OR C1.INPUT > 5 

サンプル コードの説明 
SELECT/C1.INPUT 
このコード 
ラインは、ユーザーによってタイプ入力された番号、または、ストリング値（この例の場合

、番号）を受け取り、その入力内容から、どのCASE/END_CASE 
ブロックが実行されるか決定します。 SELECT / END_SELECT 
ペアが、このコード一覧全体を囲んでいることに注意して下さい。すべてのCASE / 

END_CASEとDEFAULT CASE / END_DEFAULT 

CASEペアが、これら2つの行間に配置される必要があります。 

END_SELECT 
これは、「SELECT/END_SELECT」ペア内のコードの終了をマークします。 

CASE/1 から CASE/5まで 
C1.INPUTの値により、CASE コードのうちの一つのブロックが実行されます。 
例えば、C1.INPUTが1と評価すると、CASE 1 
コードブロックが実行され、CIR1を測定します。 2と評価すると、CASE 2 
コードブロックが実行され、CIR2を測定する、といった具合です。 

END_CASE 
この行は、特定のCASEコード ブロックを終了します。 

DEFAULT CASE 
C1.INPUTの値が、あらかじめ定義されたCASEステートメントのいずれとも一致しない場

合には（値が、1から5までの数字でない場合）、DEFAULT CASEコード 
ブロックが実行されます。 
この場合、ループから退出することを知らせるメッセージが表示されます。 

Do / UNTIL ループが、コード 
サンプル全体を囲んでいることに注意して下さい。これを用いると、ユーザーは、CASEス
テートメントが認識しない文字を選択するまで、COMMENT/INPUTラインから作成されたメ

ニューから選択し続けることができます。 

CASE / END CASE 

挿入 | フローコントロール コマンド | コントロール ペア | CASE / END CASEメニュー 
オプションを選ぶと、測定ルーチンに、条件ブロックを付け加えることができます。CASEステー

トメント用の入力式が、それに対応するSELECTコマンドの入力式と等しい数値を出す場合、CAS
EコマンドとEND 
CASEコマンドの間にある項目が実行されます。それ以外の場合には、そのブロックのステートメ

ントは省略されます。CASE / END CASEステートメント 
ブロックは、SELECTコマンド、あるいは、その前のCASE / END CASEブロックのEND 
CASEコマンドの直後に置かれる必要があります。また、PC-
DMISは、単一の場合ステートメント上で、複数の式を比較することはできません。 

CASE / END 
CASEステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります: 
 

CASE/式 
END_CASE/ 

「CASE / END CASE」オプションを挿入するには: 

1. 編集ウィンドウ内のご希望の位置に、カーソルを配置して下さい。上記に述べられた、位

置付けに関する必要条件に注意して下さい。 

2. メニューバーからCase / End Caseを選択します。CASE / End Case 
ステートメントが編集ウィンドウ内に表示されます。 

Default Case / End Default Case 

挿入 | フローコントロール コマンド | コントロール ペア | Default Case / End Default 
Caseメニュー 
オプションを選ぶと、測定ルーチンに、条件ブロックを付け加えることができます。「SELECT / 
END SELECT」ブロック内の、先行する「CASE / END 
CASE」ブロックの、その他すべての式が、偽と出た場合に、「DEFAULT CASE」と「END 
DEFAULT CASE」の間の項目は、実行されます。「SELECT / END 
SELECT」ブロック内では、ひとつの「DEFAULT CASE /END DEFAULT 
CASE」ブロックのみが存在できます。「DEFAULT CASE /END DEFAULT 
CASE」ブロックは、「SELECT / END SELECT」ブロック内のすべての「CASE / END 
CASE」ブロックの後に置かれる必要があります。 

「DEFAULT CASE / END DEFAULT 
CASE」ステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになりま

す: 

DEFAULT CASE/  
END_DEFAULT_CASE/ 

「DEFAULT CASE / END DEFAULT CASE」コマンドを挿入するには: 

1. 上記に記述された位置付けの制限に注意して、編集ウィンドウ内のご希望の位置に、カー

ソルを配置して下さい。 

2. メニューバーからDefault Case / End Default Caseを選択します。DEFAULT CASE / End 
DEFAULT CASE ステートメントが編集ウィンドウ内に表示されます。 

汎用ループの使用 

 
ループ·パラメータダイアログボックス 

挿入|フロー制御コマンド|ループメニューオプションは、 
ループ·パラメータダイアログボックスが表示されます。オフセットのどれで、あるいはどれもな

しでも測定ルーチン(あるいは測定ルーチンの一部)を繰り返すLOOPコマンドを作成するには、こ

のダイアログ・ボックスを使用することができます。LOOPコマンドは測定ルーチンのどこにでも

追加できますが、この機能はルーチンの始めと終わりで最も役に立ちます。 

開始番号 - このボックスは、一連のパーツのなかの開始位置番号をPC-DMISに指示します。 

例: 
10個のパーツを持ち、位置番号5から開始したい場合、パーツの合計数として10を、そして

、開始位置として5を入力します。 

終了番号 - このボックスは、測定ルーチンによって何回ループするべきであるかをPC-
DMISに伝えます。この数は、通常、取付具が 

x（y、または、z）方向に持つパーツの総数（または、パーツのパターン数）と同じ数です。PC-
DMISはまた、開始のパーツ（パターン）番号についても尋ねます。 

例: 
x（y、または、z）方向に10個のパーツを持ち、位置番号5から始動したい、とします。終了

番号ボックスには、 10を入力します。開始番号ボックスには、 5を入力します。 

スキップ数 - ループでは、 PC- 
DMIS測定ルーチンを何回表示された数を繰り返します。スキップ番号ボックスを使用すると指定

した増分をスキップすることができます。このボックスを使用して最初の増分をスキップすること

はできませんが、開始番号を2に変更することで同じ効果でスキップすることができます。 

例: 
ループの3番目ごとの増分を省略するよう、パラメータを設定することが可能です。数字3が
指示されると、PC-
DMISは最初と2番目のパーツを測定しますが、その後、省略して、4番目のパーツへと進み

ます。 

オフセット - オフセット エリアには、以下のボックスがあります: 

X 軸、 Y 軸、 Z 軸 - 
これらのボックスは、パーツ間、あるいは、同一パーツ上のパターン間のX、Y、または、Zオフセ

ットをセットアップします。これらは、ループが実行される度に、入力された距離によって、パー

ツをオフセットします。最初のオフセットは、パーツの原点に基づいています。 

角度 - 
このボックスは、パーツ間、あるいは、同一パーツ上のパターン間の角オフセットをセットアップ

します。最初のオフセットは、パーツの原点に基づいています。PC-
DMISは、ループが実行される度に、角度によって、パーツをオフセットします。 

ループ内にパーツ配置コマンドがあり、そのループがオフセットを使用している場合、そのパーツ

配置用の軸線すべてを定義しなければなりません。それに加え、ループ内のパーツ配置は、ループ

内で測定されるフィーチャーを使用しなければなりません。 

ループのID - このチェックボックスは、 PC- DMIS 
、ループの増分と一致する角括弧内の値として（ループ内）要素IDを増分するか否かを判定します

。 

例: 
CIR1は、最初のループ上でCIR1[1]となり、2番目のループ上でCIR1[2]となり、以降それに

続きます。 

統計データベース内の要素ID 

ユーザが、ループIDのチェックボックスを選択すれば、いくつかのケースでは、データベー

スに統計データを送信する時に、 PC- 
DMISは、データベースにこれらのループIDを表示しない場合があります。 

手順は以下のとおり: 

ループブロックに、STATS/ON コマンド、及び、STATS/UPDATE 
コマンドがある場合、ループIDはそのデータベース内に表示されません。 

ループブロックの外にSTATS/ON 
コマンド、及びループブロックの中にSTATS/UPDATE 
コマンドがある場合、ループIDはそのデータベース内に表示されません。 

ループブロックの外にSTATS/ON 
コマンド、及びループブロックの外にSTATS/UPDATE 
コマンドがある場合、ループIDはそのデータベース内に表示されません。 

ループブロックの中にSTATS/ON 
コマンド、及びループブロックの外にSTATS/UPDATE 
コマンドがある場合、ループIDはそのデータベース内に表示されません。 

  

変数ID - 
このボックスを用いると、ループのその時点での反復（または、特定されたループ数のうちのその

時点でのループ）を追跡するために使用する変数名を定義することができます。測定ルーチン実行

中、この変数は、ループのその時点での反復数を同じ数を表示します。 

End Loop - 
このボタンを押すと、ループのプロセスが終了します。編集ウィンドウ内で、「LOOP/START」コ

マンドの後には「LOOP/END」コマンドが続く必要があります。 

  

ループの目的 

ループ オプションには、主な目的が3つあります: 

• パーツの格子を持つ、複数パーツ用の取付具があるとします。その取付具は、横列間の間

隔として一貫した空間を必要とします。位置変換 / 
回転オフセットを用いると、パーツの格子内で、ひとつのパーツから次のパーツへと指示

することができます。 

• パーツをひとつ持つ取付具があり、ルーチンの各ループの前に、取付具の持つパーツを新

規のものに交換したい、とします。新規のパーツに置き換える時には、COMMENTコマン

ドを用いてCMMを停止することができます。そのコマンドは、ループの冒頭、または、末

尾のいずれにも置くことが可能です。 

• ループ 
オプションを使用して、同一のパーツの別の部分を測定するために、測定ルーチンを回転

することができます。例えば、パーツ上で10回繰り返されている、複雑な穴のパターンを

測定するための測定ルーチンを作成するとします。ユーザの測定ルーチンは、ホールパタ

ーンの一つを測定する必要があります。そのパターンの他の9回の繰り返し部分を測定する

ために、ループ 
オプションを使用して、この測定ルーチンをオフセットすることができます。 

付記: 
ループ内部でパーツ配置を使用する場合、前回保存したアラインメントを常に呼び出すかわりに、

PC-DMIS3.6以降では、ALIGNMENT/STARTコマンド 
ラインのアクティブなアラインメントを使用することができます。「アラインメントの作成と使用

」の章にある「ループ内でアラインメントを使用」の項目を参照して下さい。 

  

  

ループの作成 

1. メニュー バーから挿入 | フローコントロール コマンド | 
ループを選択すると、ループ用パラメータ ダイアログ ボックス が表示されます。 

2. ボックス内で、必要な変更を行って下さい。 

3. 必要に応じてパラメータ（例えば、XYまたはZ、角の数、オフセットをスキップして、番

号を開始、パーツの数など）を選択します。 

4. 編集ウィンドウ内の、ループ開始予定の位置に、カーソルを配置して下さい。 

5. OKボタンをクリックします。 

ループ用の編集ウィンドウのコマンドラインは、以下のようです: 

VARNAME = LOOP/START, ID = Y/N, NUMBER = 0, START = 1, SKIP = , 
OFFSET: XAXIS = 0, YAXIS = 0, ZAXIS = 0, ANGLE = 0 

注記: ループの手続きを完了するには、LOOP/ENDコマンドで完成しなければなりません。PC- 
DMISはLOOP / STARTおよびLOOP / 

ENDコマンドによって包含されたすべての編集ウィンドウコマンドをループします。これらの方法

のうちの1つにこのコマンドを挿入することができます: 
編集ウインドウで、LOOP/ENDをタイプし、「挿入 |フロー制御コマンド| End 
Loop」メニューアイテムを選択するか、または「ループのパラメーター」ダイアログ・ボックス

から、「End Loop」ボタンをクリックします。 

  

ラベルの作成 

実行フローは、特定の条件に基づいて送信される場所を制御するためにGOTOまたはIF_GOTOコマ

ンドでLABELコマンドを使用することができます。ラベル名はスペースを持つことができず、230
文字を超過してはなりません。PC-DMISはすべて頭文字を使用して、ラベル名を表示します。 

LABELコマンドを作成するためには、下記のうちの1つを行ってください: 

コマンドを入力します。 

1. 編集ウインドウのブランク行にカーソルを置いてください。 

2. LABELをタイプし、次にタブ・キーを押します。 

3. 強調されたエリアで、新規ラベル名をタイプしてください。 

または、 

コマンドを選択する 

1. 編集ウインドウにカーソルを置いてください。 

2. メニューバーから、挿入| フロー制御コマンド | 
ラベルを選び、ラベル名の編集ダイアログボックスにアクセスします。 

3. 新規ラベル名ボックスに、ラベルのIDを入力して下さい。 

 
「ラベル名編集」ダイアログ ボックス 

4. OKをクリックすると編集ウインドウの次の可能な位置にラベル名が挿入されます。 

編集ウィンドウのコマンドモードのLABELコマンドは次のものを読みます:  
ID = LABEL/ 

ここでIDは、指定されたラベルの名前を表わします。 

ラベルを備えたGOTOコマンドの使用に関する情報については、「GOTOを使用するラベルにジャ

ンプする」トピックを参照してください。 

ラベルを備えた 

IF_GOTOコマンドの使用に関する情報については、「GOTOを使用するラベルにジャンプする」ト

ピックを参照してください。 

GOTOを使用してラベルにジャンプ 

GOTOコマンドの作成により特定のラベルにジャンプするように測定ルーチンに伝えることができ

ます。 

GOTOコマンドを作成するためには、下記のうちの1つを行ってください: 

コマンドを入力します。 

1. 編集ウインドウのブランク行にカーソルを置いてください。 

2. GOTOをタイプし、次にタブ・キーを押します。 

3. ジャンプ先のラベル名を入力します。 

または、 

コマンドを選択する 

1. 編集ウインドウにカーソルを置いてください。 

2. メニューバーから、挿入 | フロー制御コマンド | GoToを選んで 
GoToダイアログボックスにアクセスします。 

「行き先」ダイアログ ボックス 

3. ラベルが測定ルーチンの中にすでに存在する場合、それらが現在のラベルボックスに現れ

ます。 

4. 現在のラベルから、目的のラベルを選択するか、GoToのラベルボックスに、直接的にラ

ベル名を入力します。 

5. OKをクリックすると編集ウインドウの次の可能な位置にGOTOコマンドが挿入されます。 

編集ウィンドウのコマンドモードのGOTOコマンドは次のものを読みます:  
GOTO/label_ID 

ここでlabel_IDは、指定されたラベルの名前を表わします。ラベル名はスペースを持つことができ

ず、230文字を超過してはなりません。 

注意: 
ラベルがまだ作成されていない場合、ラベル名は編集ウィンドウのコマンドモードの赤で強調表示

されます。また、そのGOTOコマンドは実行の間に無視されます。 

 
「If 式」ダイアログ ボックス 



フローコントロールを使っての分枝 

22 

「挿入 | フロー制御コマンド | If Goto」オプションを使用しると、測定ルーチンの内に「IF 
GOTO」文をつくるために「If 
Expression」ダイアログボックスを使うことができます。測定ルーチンの実行中に、PC-
DMISがIF 
GOTOステートメントに出会うと、特定の式がゼロ以外の値を出した場合に、ルーチンのフローは

ラベルIDの所在位置に指します。  

IF 
GOTOステートメント用の、編集ウィンドウ内のコマンドラインは、以下のようになります

: 
 

IF_GOTO/式, GOTO=Label 

式 - 
式をクリックすると、式ビルダが開きます。式ビルダを用いると、測定ルーチンに必要な様々な式

を作成できます。式が作成されたら、それは[式]ボックスに表示されます。式の作成に関する情報

については、「式及び変数の使用」の章を参照して下さい。 

ラベル - 
ラベルがクリックされると、Gotoのダイアログボックスが表示されます。Gotoダイアログボック

スで、定義された式が満たされたときにルーチンの流れはどのラベルへ移動するかを選択すること

ができます。選択されたラベルは、 
[ラベル]ボックスに表示されます。あるいは、ラベル名を知っていれば、ボックスにそれをタイプ

することができます。ラベルについての詳細は、「ラベルにジャンプする」を参照してください。 

エラーにあたっての分枝 

 
「エラーにあたって」ダイアログ ボックス 

挿入| フロー制御コマンド| 
オン・エラー」コマンドは、「オン・エラー」ダイアログ・ボックスを開きます。ユーザは機械の

エラーが起こったときに実行するアクションをPC- 
DMISに伝えるために、このダイアログボックスを使用することができます。 

エラータイプ - PC-DMISは、以下のエラー状況を追跡します: 

• 予期しないﾌﾟﾛｰﾌﾞﾋｯﾄ 

• 省略されたプローブ ヒット 

• 反射装置が発見されず（PC-DMISポータブルにおいて、追跡機と共に使用） 

• レーザーのエラー（レーザー構成を使用している場合、「ONERRORを使用したレーザー

プローブエラーの処理」を参照してください） 

エラーモード - これらのエラー状況のそれぞれにおいて、以下の行動をとることが可能です: 

• GoTo Label - 
測定ルーチン・フローは定義されたラベルに移ります。「ラベルの使用」を参照してくだ

さい。 

• 変数の設定: 変数の値を1に設定します。 

• オフ: PC-DMISは何もしません。 

• スキップコマンド: 
測定ルーチンはその時点でのコマンドを省略し、測定ルーチン内にマークされた次のコマ

ンドに移動します。 

デフォルトでは、すべての測定ルーチンがオフ（何もしない）に設定された両種類のエラーのアク

ションを開始します。各エラーの行動モードは、測定ルーチンを通して変更することができます。 

ラベルID - ユーザはまだ存在していないラベルへの参照を入力できます。 

現在のラベル - 測定ルーチン内のすべてのラベルを一覧表示します。 

例: 実行の間、PC-DMISはエラー時 / 予定外のヒット / 

ラベルへジャンプコマンドに遭遇すると、測定ルーチンのポイント後に起こるどんな予期していな

かったヒットでも、実行は指定されたラベルにジャンプします。変数の値の設定への動作は指定さ

れたエラータイプが起こるとすぐに、変数が設定されることを引き起こします。測定ルーチンの新

しい点にジャンプさせるIFステートメントを使用してこの変数の値をテストできます。 

「On Error」コマンドを使用するには: 

1. 挿入| フロー制御コマンド| 
オン・エラー」を選択して、「オン・エラー」ダイアログ・ボックスを開きます。 

2. エラータイプのリストから、分岐の条件として、エラーの種類を選択します。 

3. 「エラー·モード」領域で、その条件が満たされた場合に行われるべきアクションを定義す

るためのオプションのいずれかを選択します。 

4. ONERRORコマンドを追加するには、[OK]ボタンをクリックしてください。キャンセルボタ

ンは、いかなる変更も行わず、On Errorダイアログボックスを閉じます。 

On Errorコマンド用の編集ウィンドウ上のコマンドラインは、以下のようになります: 

ONERROR/<type>,<mode> 

ここで、<タイプ>はエラーの種類で、<モード>は取られたエラーモードの類別です。 

「On Error」支援のインターフェース 

すべてのインターフェースが、「On 
Error」コマンドに対応しているわけではありません。お客様のインターフェースが対応している

か、確認するには、以下の表をご覧下さい。 

• インターフェースが一覧に表示されている場合、小さな黒いボックスはそのエラーのタイ

プがそのインターフェースに対応していることを示します。 

• お客様のインターフェースが一覧に表示されていない場合、ON 
ERRORコマンドを使用することはできません。 

対応のフィーチャー 予期しないﾌﾟﾛｰﾌﾞﾋｯﾄ 
省略されたプローブ 

ヒット 
反射装置が発見されず 

B&S 標準 ■ ■ - 

DEA   ■ - 

ELM ■ ■ - 

フェデラル/ルノー ■ ■ - 

ヨハンセン ■ ■ - 

ライカ追跡機     ■ 

ライツ ■ ■ - 

LK Direct (別名 
LKRS232) 

■ ■ - 

LKﾄﾞﾗｲﾊﾞ ■ ■ - 

アラインメント ■ ■ - 

ミツトヨ ブライト ■ ■ - 

ミツトヨ ■ ■ - 

Mora ■ ■ - 

Omnitech ■ ■ - 

レ二ショー ■ ■ - 

シャープ ■ ■ - 

Sheffield ■ ■ - 

Wenzel ■ ■ - 

Zeiss ■ ■ - 

  

サブルーチンでの分枝 

サブルーチンは、お客様の測定ルーチン内や、外部測定ルーチン内の、コードのブロックであり、

通常繰り返して参照され、より簡潔なプログラム作成が可能になります。PC-
DMISでは、サブルーチン内の「引数」（または、ローカル変数）に情報を渡すことができます。

サブルーチンに引き渡すことが可能な引数のタイプは、数値、変数、テキスト文字列と要素名です

。 

ヒント: サブルーチン コマンド 
ブロックは、SUBROUTINEコマンドとENDSUBコマンドの間にあります。 

測定ルーチン内にサブルーチンを作成した後、それを、その時点での測定ルーチンや別の測定ルー

チンから「呼び出す」ことができ、それを行うと、測定ルーチンの実行フローが特定のサブルーチ

ンに入り込み、サブルーチン コマンド 
ブロック内のコマンドを実行します。測定ルーチンのフローは、その後、呼び出しステートメント

の直後に続くステートメントへと戻ります。 

ヒント: サブルーチンは、CALLSUBコマンドを用いて、呼び出されます。 

外部サブルーチン 

外部サブルーチン、または、呼び出し測定ルーチンの外部にある測定ルーチンのサブルーチンは、

呼び出し測定ルーチンにある要素、変数、または、アラインメントにアクセスすることはできませ

ん。サブルーチンは、それ自身の測定ルーチン内の項目にはアクセスできます。外部測定ルーチン

と呼び出し測定ルーチンは、同じ測定単位を使用する必要があります。 

サブルーチンのネスト化 

他のサブルーチン内でサブルーチンをネストできます。引数とネストされるサブルーチンの数はメ

モリの空き容量のみによって制限されます。 

  

  

サブルーチンの新規作成 

SUBROUTINE入力によるサブルーチンの作成 

編集ウィンドウ内にSUBROUTINEと入力し、その後、タブキーを押すことにより、このコマンド

を挿入することが可能です。いったんそのコマンドが挿入されると、そのサブルーチン名と引数を

特定する必要があります。詳細は以下のサブルーチン構文と例を見てください。 

ENDSUBコマンドを入力することとTABを押すことで コマンド 
ブロックを終了します。このコマンドブロックの中に置任意の編集ウィンドウのコマンドも、サブ

ルーチンの一部であると考えられて、サブルーチンが呼ばれるとき、実行されます。 

サブルーチン メニュー項目を使用したサブルーチンの作成 

1. サブメニューから挿入 | フローコントロール コマンド | 
サブルーチンオプションを選択して下さい。これによって、[サブルーチン作成]ダイアロ

グ ボックスが現れます。このダイアログ 
ボックスに関する情報については「サブルーチン作成ダイアログ 
ボックスについて」の項目を参照して下さい。 

「サブルーチン作成」ダイアログ ボックス 

   
2. 名称ボックス内に、サブルーチンの名称を入力して下さい。名前の長さは180文字以内に

制限されています。181文字以上入力すると、「行が長すぎます」のエラーメッセージが

現れます。 

OKボタンを選択すると、その名前が最初の180文字に自動的に短縮され、サブルーチン作

成ダイアログボックスが閉じます。 

3. サブルーチンが引数(サブルーチン内に引き渡された情報のプレース 
ホルダ)を使用する場合、[引数を追加]ボタンをクリックして、それらをひとつずつ追加し

ます。引数の編集ダイアログ ボックスが現れます。このダイアログ 
ボックスに関する情報については、「引数の編集ダイアログ 
ボックスについて」の項目を参照して下さい。 

 
引数編集ダイアログ ボックス 

4. 名前ボックス内に、引数の名前を入力して下さい。 

5. 数値ボックス内に、引数のデフォルト設定値を入力して下さい。CALLSUBステートメント

からサブルーチンに何の値も渡されない場合、サブルーチンはそのデフォルト設定値を使

用します。有効な引数の値は、数値、変数、文字列、および要素名です。 

6. 引数に説明を付けたい場合は、説明ボックス内に入力して下さい。 

7. 引数を作成するために、引数の編集ダイアログ ボックスのOKをクリックして下さい。 

8. サブルーチンに含めたい引数それぞれについて、ステップ3から7までを繰り返して下さい

。 

9. サブルーチン作成を完了するには、サブルーチン作成ダイアログ 
ボックスのOKボタンをクリックして下さい。このサブルーチンが、あらかじめ定義された

引数すべてと共に、編集ウィンドウ内に表示されます。 

10. 挿入 | フローコントロール コマンド | 
サブルーチン終了メニューオプションを選択して、サブルーチンを終了します。編集ウィ

ンドウに「ENDSUB/」コマンドが配置され、 サブルーチンコマンド 
ブロックを終了します。その他の測定ルーチンのコマンドをサブルーチンに含めたい場合

、このサブルーチンのコマンドブロック内でENDSUBコマンドの前に加える必要がありま

す。 

  

サブルーチン コマンド ブロック用のシンタックス 

サブルーチンコマンドブロックのサンプル用の編集ウィンドウのコマンドライン 
シンタックスは、以下のようになります:  

SUBROUTINE/<名前>, 

<A1> = <引数1> : <説明>, 

<A2> = <引数2> : <説明>, 
= 

<コマンド> 
ENDSUB/ 

SUBROUTINE/ は、サブルーチンのコマンド 
ブロックを開始するために使用されるコマンドです。 

<名前> = 
サブルーチン名。名前の長さは256文字以内に制限されています。257文字以上を入力した場合、

名前は自動的に256文字に短縮されます。 

<A1> = 
サブルーチン内で使用される最初の引数（あるいは、ローカル変数）。この変数は、一般的に、サ

ブルーチンの外側からはアクセスできません。 

<A2>= サブルーチン内で使用される2番目 
の引数。この変数は、一般的に、サブルーチンの外側からはアクセスできません。必要に応じて、

追加の引数を追加できます。 

<引数1> = 最初の引数のデフォルト値。 

<引数2> = 2番目の引数のデフォルト値。 

<説明> = 引数の内容説明。 

<コマンド> = 
その他の編集ウィンドウ用のコマンドを、必要に応じて、引数の後、ENDSUBコマンドの前に挿入

することが可能です。 

ENDSUB/は、サブルーチンのコマンド ブロックを終了するために使用されるコマンドです。 

サブルーチン コマンド ブロックの例 

例えば、オペレーターのデータを受け取り、それをレポートに表示する、サブルーチンの完了した

ものは、以下のようになります: 

SUBROUTINE/GET_OPERATOR_INFO, 

   OPNAME = <オペレーター> : OPERATOR NAME, 

   SHIFT = <シフト> : SHIFT TIME, 

= 

COMMENT/REPT,OPNAME 

COMMENT/REPT,SHIFT 

ｻﾌﾞﾙｰﾁﾝの終了/ 

サブルーチン作成ダイアログ ボックスの理解 

サブルーチンの作成ダイアログボックスにアクセスするには、挿入 
|フロー制御コマンド|サブルーチンを選択しします。 

「サブルーチン作成」ダイアログ ボックス 

下記のオプションがサブルーチンの作成ダイアログボックスで使用できます。 

名称 
ボックスはサブルーチンを定義します。これは後でサブルーチンを呼び出したときに使用する名称

であるため、一つの測定ルーチンに複数のサブルーチンがある場合はそれぞれ一意の名称でなくて

はなりません。 

名前の長さは180文字以内に制限されています。名称ボックスに181文字以上入力すると、

「行が長すぎます」のエラーメッセージが現れます。 

OKボタンを選択すると、その名前が最初の180文字に自動的に短縮され、サブルーチン作

成ダイアログボックスが閉じます。 

引数の番号この一覧には、作成中のサブルーチンの引数が表示されます。このエリアでは、引数が

以下の形式で表示されます: 

<名称> =<数値> :<内容説明> 

例えば、引数うちのひとつが「直径」という名称で、そのデフォルト設定値が3である場合

、この一覧内で、引数は以下のように表示されます: 

直径 = 3 : 穴の直径 

CALLSUBコマンドから渡される、別の値がない時には、PC-
DMISは常にデフォルト設定値を使用します。 

引数を編集するには、変更したい引数をダブルクリックします。引数の編集ダイアログ 
ボックスが開き、それを用いて、必要な変更を行うことができます。このダイアログ 
ボックスに関する情報については、「引数の編集ダイアログ 
ボックスについて」の項目を参照して下さい。 

引数を追加 - 
このボタンは、サブルーチンに新規の引数を追加します。このボタンをクリックすると引数を編集

ダイアログ ボックスが現れます。このダイアログ 
ボックスに関する情報については、「引数の編集ダイアログ 
ボックスについて」の項目を参照して下さい。 

引数を削除 - 
このボタンを用いると、サブルーチンから引数を削除することができます。一覧からその引数を選

択し、引数を削除ボタンをクリックして、それを削除します。 

既存のサブルーチンの編集 

既存のサブルーチンを編集するには、常に、編集ウィンドウのコマンド 
モードを用いて、直接そのサブルーチンを編集することが可能です。その代わりに、編集ウィンド

ウ内のサブルーチン上にカーソルを配置し、F9を押して、 [サブルーチン作成] ダイアログ 
ボックスにアクセスすることも可能です。これによって、[サブルーチン作成]ダイアログ 
ボックスが現れます。このダイアログ ボックスに関する情報については 
「サブルーチン作成ダイアログ ボックスの理解」の項目を参照して下さい。 

  

引数編集ダイアログ ボックスの理解 

 
引数編集ダイアログ ボックス 

サブルーチン作成ダイアログ ボックス(挿入 | フロー制御コマンド| 
サブルーチン)、または、サブルーチンの呼び出しダイアログ ボックス(挿入 | フロー制御コマンド| 
サブルーチンの呼び出し)内で、引数の作成、または、編集を選択するたびに、引数編集ダイアロ

グ ボックスが表示されます。 

引数編集ダイアログ ボックスは、以下の2つの状況下で使用が可能です: 

• SUBROUTINEコマンド 
ブロック内で、サブルーチンの引数、及び、それらのデフォルト設定値を定義するため。 

• CALLSUBコマンドからサブルーチン内に渡される数値を定義するには、 

次のオプションは引数編集ダイアログ・ボックスにおいて利用可能です: 

名前 - このボックスには、作成か編集している引数の名前を定義します。 

値 - このボックスには、引数の値を定義します。 

SUBROUTINEコマンドを作成、または、編集する場合、これが、CALLSUBステートメント

からサブルーチンに何の値も渡されない場合に使用されるデフォルト設定値となります。 

CALLSUBコマンドを作成、または、編集する場合、これが、サブルーチンに渡される値と

なります。 

その場合の有効な値は: 

• 数値 

• 変数 

• テキスト文字列- 
テキスト文字列は、二重引用符で囲まれていなければなりません。 

• フィーチャー名 - 例えば、{F1} 
のように、フィーチャー名は、波状カッコで囲まれていなければなりません。 

説明 - 
サブルーチンの引数の説明を定義します。この内容説明は、編集ウィンドウ内のSUBROUTINEコマ

ンド ブロックにおいて、引数の横に表示されます。 

  

サブルーチン呼び出し 

サブルーチンを呼び出すには、CALLSUBコマンドをお客様の測定ルーチン内に挿入する必要があ

り、それによって、その時点での測定ルーチンから既存のサブルーチンを呼び出すか、または、外

部測定ルーチンからサブルーチンを呼び出します。 

CALLSUB入力によるサブルーチンの呼び出し 

編集ウィンドウ内でCALLSUBと入力して、このコマンドを挿入することができるうえ、その後、

編集ウィンドウにコマンドを予定の場所で表示するようにキーボードでTab 
キーを押します。いったんそのコマンドが挿入されると、そのサブルーチン名、外部測定ルーチン

内にある場合その所在位置、そして、利用可能な引数に渡す数値を特定する必要があります。渡す

引数の例については、「引数をサブルーチンへ引渡し」を参照して下さい。 

Call Subメニュー項目を用いたサブルーチンの呼び出し 

1. サブメニューから[挿入 | フローコントロール コマンド | 
サブルーチン呼び出し]オプションを選択して下さい。サブルーチン呼び出しダイアログ 
ボックスが開きます。このダイアログ 
ボックスに関する情報については「サブルーチン呼び出しダイアログ 
ボックスについて」を参照して下さい。 

 
「サブルーチン呼び出し」ダイアログ ボックス 

2. サブルーチン選択ボタンをクリックします。サブルーチン選択ダイアログ 
ボックスが開きます。 

「サブルーチン選択」ダイアログ ボックス 

3. ユーザー ディレクトリチェック ボックス、現在のディレクトリチェック 
ボックス、またはその両方を選択して下さい。サブルーチンが含まれる測定ルーチンが、

サブルーチン検索で指定されたディレクトリ内にある場合、ユーザー 
ディレクトリチェック 
ボックスを選択して下さい。その時点でのディレクトリ内にある場合には、現在のディレ

クトリチェック ボックスを選択して下さい。PC-
DMISは、選択可能な測定ルーチンすべてを一覧表示します。 

4. ご希望のサブルーチンを含む測定ルーチンを選択して下さい。選択された測定ルーチンに

関連したサブルーチンすべてがサブルーチン名ボックス内に表示されます。 

5. 呼び出したいサブルーチンを選択して下さい。 

6. OK 
ボタンをクリックして下さい。呼び出し予定のサブルーチンについての情報が、[サブルー

チン呼び出し]ダイアログ 
ボックスの[名称]ボックスとファイル]ボックス内に表示されます。 

7. そのサブルーチン内に情報を渡す場合、引数追加ボタンをクリックし、引数編集ダイアロ

グ ボックスを使用して、渡す引数と数値を定義して下さい。このダイアログ 
ボックスに関する情報については、「引数編集ダイアログ 
ボックスについて」を参照して下さい。渡す引数の例については、「引数をサブルーチン

へ引渡し」を参照して下さい。 

8. 再びOKボタンをクリックして下さい。CALLSUBコマンドが編集ウィンドウ内の選択され

た場所に表示されます。 

CALLSUBコマンド用のシンタックス 

サブルーチン呼び出し用の編集ウィンドウのコマンドライン シンタックスは、以下のようです: 

CS1        =CALLSUB/<名称>, <ファイル>:<引数1>,<引数2>, 

CS1 = CALLSUBコマンドに与えられた、ラベルID。 

<名称> = 呼び出されるサブルーチン名。 

<ファイル> = 
呼び出すサブルーチンを含む、測定ルーチンへの完全な経路。このフィールドが空白である場合、

PC-DMISは、その時点での測定ルーチン内で、そのサブルーチンを探します。 

<引数1> = 
サブルーチン内の最初の引数に渡される値。このフィールドが空白である場合、代わりに、サブル

ーチン内の最初の引数用に定義されたデフォルト設定値が使用されます。 

<引数1> = 
サブルーチン内の２番目の引数に渡される値。このフィールドが空白である場合、代わりに、サブ

ルーチン内の２番目の引数用に定義されたデフォルト設定値が使用されます。この構文サンプルは

、2つの引数だけを示します。サブルーチンで必要な場合、他の引数を引き渡せます。 

注記: 
CALLSUBコマンド内では、そのサブルーチン用に作られた全オブジェクトへの、一連のポインタ

を保持すべきで、それによって、後にサブルーチンIDを用いて、簡単にそれらを参照することがで

きます。ポインターに関するより詳しい説明については「式および変数の使用」の章にある「ポイ

ンタ」を参照して下さい。 

  
CALLSUBコマンドの例 
CS1 
=CALLSUB/GET_OPERATOR_INFO,D:\MEASROUTINES\V42SUBROUTINETEST.PRG:V1,V2,, 

この、CALLSUBコマンドの例、CS1は、D:\MEASROUTINES\ 
ディレクトリ内にある、測定ルーチンV42SUBROUTINETEST.PRG内に位置する、GET_OPERAT
OR_INFOという名称のサブルーチンを呼び出します。 

これは、2つの数値ーこの場合、変数V1とV2ーをサブルーチン内に渡します。 

  

  

サブルーチン呼び出しダイアログ ボックスの理解 

「サブルーチン呼び出し」ダイアログ ボックス 

次にサブルーチンの呼び出しダイアログ ボックス内で利用可能なオプションを説明します。 

名前 ボックス 

   

名称ボックスには、サブルーチン選択...ボタン使用後に選択されたサブルーチンの名

称が含まれます。 

名前の長さは180文字以内に制限されています。名称ボックスに181文字以上入力す

ると、「行が長すぎます」のエラーメッセージが現れます。 

OKボタンを選択すると、その名前が最初の180文字に自動的に短縮され、サブルーチ

ン呼び出しダイアログボックスが閉じます。 

ファイル ボックス 

ファイルボックスには、呼び出されたサブルーチン 
ファイルのディレクトリパスが表示されています。 

値 ボックス 

   

数値ボックスには、そのサブルーチンに関連した各引数の値の一覧があります。サブ

ルーチンが実行される時、これらの値はサブルーチン内に渡されます。 

これらの値を編集するには、変更したい値をダブルクリックします。引数の編集ダイ

アログ ボックスが現れます。このダイアログ 
ボックスに関する情報については、「引数の編集ダイアログ 
ボックスについて」の項目を参照して下さい。 

サブルーチンの選択 ボタン 

   

サブルーチン選択ボタンを押すとサブルーチン選択ダイアログ 
ボックスが現れます。 

 
「サブルーチン選択」ダイアログ ボックス  

このダイアログ ボックスでは、ユーザー 
ディレクトリまたはその時点でのディレクトリ内を検索し、以前に作成されたサブル

ーチンを呼び出すことができます。サブルーチン選択ダイアログ 
ボックスに関する情報については、「サブルーチンの呼び出し」を参照して下さい。 

引数の追加 ボタン 

 

このボタンは、サブルーチン用の引数に渡す値を定義します。 

引数の削除 ボタン 

   

このボタンを押すと、値ボックス内の引数を削除することができます。表示された値

を選択し、引数を削除ボタンをクリックして下さい。その値に関連した引数が削除さ

れます。 

  

「サブルーチン呼び出し」ダイアログ ボックスを用いて、新規の引数を追加するには: 

  

サブルーチン内に渡すための、新規の引数をCALLSUBコマンドに付け加えるには: 

1. サブルーチン呼び出しダイアログ 
ボックスで引数追加ボタンをクリックして引数編集ダイアログ ボックスを開きます。 

2. 値ボックスをクリックして下さい。 

3. 値ボックス内では、引数の値を入力できます。 

4. [OK] をクリックします。 

  

「サブルーチン呼び出し」ダイアログ ボックスを用いて、既存の引数を編集するには: 

  

CALLSUBコマンド内の、既存の引数を編集するには: 

1. サブルーチン呼び出しダイアログ 
ボックスで、変更予定の引数の値をダブルクリックして下さい。数値ボックスが現れ、呼

び出されたサブルーチンのデフォルト値を表示します。 

2. 新しい値を入力します。 

3. OKボタンをクリックします。 

サブルーチン用の新規の引数を編集、または、作成する方法に関する、より詳しい説明については

、「新しいサブルーチンの作成」、及び、「既存のサブルーチンの編集」を参照して下さい。 

  

Call Subroutineコマンドから引数を削除するには： 

1. CALLSUBコマンド上に、カーソルを配置して下さい。 

2. サブルーチン呼び出しダイアログ ボックスにアクセスするために、F9を押して下さい。 

3. 引数のリストから、 1つ以上の引数を選択します。 

4. 引数削除ボタンをクリックして下さい。 

5. [OK] をクリックします。 

また、編集ウィンドウ内のテキストにある引数を直接削除することも可能です。このためには次の

操作を行います: 

1. PC-DMISをコマンド モードにして下さい。 

2. CALLSUBコマンド上にカーソルを配置し、ご希望の引数がハイライト表示されるまで、タ

ブ キーを押して下さい。 

3. 引数を削除するには、文字「del」と入力します。Delete または 
Backspaceキーを押すことは本当に引数を削除しなくて、それらは、単に空の引数に引数

を変更します。 

  

複数アームモードでのCALLSUBステートメントの使用 

CALLSUB ステートメントをアーム１に割り当てた場合、 
サブルーチンが呼び出されたときにそのサブルーチン内のコマンドすべてが、アーム1に割り当て

られます。 

CALLSUB ステートメントをアーム2に割り当てた場合、 
サブルーチンが呼び出されたときにそのサブルーチン内のコマンドすべてが、アーム2に割り当て

られます。 

CALLSUB ステートメントを両アーム用にマークした場合、PC-
DMISは、そのサブルーチンのマークをオリジナル設定のままにしておきます。 

サブルーチンがMOVE/SYNC コマンドを含み、 CALLSUB 
ステートメントがアーム1またはアーム2に割り当てた場合、PC-
DMISは実行時に、これが無効であり、サブルーチンが呼び出されないことを示すエラー 
メッセージを表示します。 

特定アームの実行のためにコマンドを割り当てる情報については、「複数アーム 
モードの使用」の章にある「アームにコマンドを設定」の項目を参照して下さい。 

  

サブルーチンの例 

一般に、渡される引数とサブルーチンの例については、以下の項目内の情報を参考にして下さい。 

• サブルーチン内に引数を引き渡し 

• サブルーチンのコードのサンプル 

  

引数をサブルーチン内に渡す 

サブルーチンに引き渡すことが可能な引数のタイプは、数値、変数、テキスト文字列と要素名です

。値を引数内に渡すには、引数編集ダイアログ 
ボックスの数値ボックス内に値をタイプ入力するか、または、コマンド 
モードにおいて、挿入されたCALLSUBコマンドに直接、値を入力して下さい。 

これを行う方法の詳細については、 「引数の編集ダイアログボックスについて」 
および「新規サブルーチンの作成」を参照してください。 

変数をサブルーチン内に渡す 

データを渡し返すことができる引数は、変数です。変数をサブルーチンへの引数として用いる時、

サブルーチン内で、それに対応する変数に起こる変化は渡し戻され、渡された側の変数の値になり

ます。 

 

渡される変数の例: 

この例は、サブルーチンへの引数として使用される変数の値が、どのように変更され、どの

ようにサブルーチンに渡し返されるかを表示します: 

変数V1には、数値6が割り当てられています: 

ASSIGN/ V1 = 6. 

サブルーチン呼び出しが、V1を最初の引数として渡します: 

CS1 =CALLSUB/MYSUB,:V1,,, 

  

そのサブルーチンは、以下のように定義されます: 

SUBROUTINE/MYSUB, 

A1 = 0 : FIRST ARGUMENT, 

= 

ASSIGN/ A1 = A1 + 1 

ENDSUB/ 

A1は最初の引数の名称であり、呼び出しが行われると、A1は、V1が呼び出し時に割り当て

られた値と同じ値、6を持ちます。 

サブルーチンには、このステートメントひとつがあります: 

ASSIGN/ A1 = A1 + 1. 

これは、A1の値を7に増加します。 

その後、サブルーチンはENDSUB/コマンドによって終了します。 

実行フローは、CALLSUBコマンドの直後に続く、ステートメントに戻ります。実行がジャ

ンプして戻ると、引数として用いられた変数、この場合はV1、は、サブルーチン内の、そ

れに対応する変数、この場合はA1、に更新されます。したがって、ここでV1は、7の値を持

ちます。その値が、サブルーチンに渡し返されます。 

数値をサブルーチン内に渡す 

引数はまた、数値を受け取ることも可能です。 

 

渡される数値の例: 

この例は、サブルーチン内で、番号の値がどのように渡されるかを表示します。最大2つの

番号を通過して、それらを合計します。 

CS1 =CALLSUB/SUM_NUMBERS,:,,, 

CS2 =CALLSUB/SUM_NUMBERS,:5,10,, 

  

SUBROUTINE/SUM_NUMBERS, 

NUM1 = 1 : FIRST NUMBER, 

NUM2 = 1 : SECOND NUMBER, 

= 

COMMENT/OPER,NO,"The sum of the first number, " + NUM1 + ", plus the 
second number, " + NUM2 + ", is: " 

,NUM1 + NUM2 

ENDSUB/ 

最初のCALLSUBコマンド（CS1）において、サブルーチン内に渡された数値はありません。

デフォルト設定値、NUM1として1、及び、NUM2として1が、かわりに使用され、その合計

は2になります。 

2番目のCALLSUBコマンド（CS2）において、2つの数値、5と10がサブルーチン内に渡され

ます。したがって、NUM1は５、そして、NUM2は10で、合計15となります。 

  

テキスト ストリングをサブルーチン内に渡す 

引数はまた、テキスト ストリングを受け取ることも可能です。テキスト 
ストリングを渡すには、英数字が、ダブル引用符内に入っていることを確認して下さい。 

 

渡される文字列の例: 

この例は、サブルーチン内で、ストリング値がどのように渡されるかを表示します。それは

、2つのストリング値を2つのパラメータ内に渡し、その後、それらをレポートに表示します

: 

CS1 =CALLSUB/GET_OPERATOR_INFO,:"BOB JONES","MORNING",, 

... 

  

SUBROUTINE/GET_OPERATOR_INFO, 

   OPNAME = <オペレーター> : OPERATOR NAME, 

   SHIFT = <シフト> : SHIFT TIME, 

COMMENT/REPT,OPNAME 

COMMENT/REPT,SHIFT 

ENDSUB/ 

  

最初の引数OPNAMEは、値「BOB 
JONES」を渡され、2番目の引数SHIFTは、「MORNING」を渡されます。その後、COMM
ENTコマンドが、渡されたストリングをインスペクション レポートに送ります。 

フィーチャー名をサブルーチン内に渡す 

要素名は中カッコで引き渡されます。たとえば; {F1} 
はF1として編集ウィンドウに指定される要素を呼ぶのに使用されます。また、要素名を通過すると

き、サブルーチンはその要素にフルアクセスできます。 

 

渡される要素名の例: 

この例は、サブルーチン内にPNT1というフィーチャー名を渡し、これによって、このサブ

ルーチンはそのフィーチャーに完全なアクセスを得ます。渡される値がない場合、F1という

デフォルト 
フィーチャー名が使用されます。このサブルーチンは、それから、新規の値について尋ね、

そのフィーチャーのXの理論値を変更します。 

CS1 =CALLSUB/CHANGE_THEO_X,:{PNT1},, 

  

SUBROUTINE/CHANGE_THEO_X, 

FEAT1 = {F1} : PASSED FEATURE NAME, 

= 

C1 =COMMENT/INPUT,NO,"PASSED FEATURE:" 

,FEAT1 

,"現在のXの理論値: " + FEAT1.TX 

,「現在のテオⅩををタイプ入力して下さい：」 

ASSIGN/FEAT1.TX = C1.INPUT 

COMMENT/OPER,NO,"要素が次の値に更新されました: " + FEAT1.TX 

ENDSUB/ 

当サブルーチンは引き渡された要素にアクセスできるので、ASSIGN/FEAT1.TX = 

C1.INPUTステートメントが、実際のPNT1要素のXの理論値を変更します。 
PNT1は、ここで、Xの理論値を恒久的に変更します。 

サブルーチンのコードのサンプル 

以下のコード 
サンプルを用いると、オペレーターは、フィーチャー測定後に、その理論的X、Y、及び、Z値の変

更を選択することができます。それ以降の実行時には、更新された理論値が使用されます。 

PNT1 =GENERIC/POINT,DEPENDENT,CARTESIAN,$ 

NOM/XYZ,<5,10,15>,$ 

MEAS/XYZ,<7,12,17>,$ 

NOM/IJK,<0,0,1>,$ 

MEAS/IJK,<0,0,1> 

C1 =COMMENT/YESNO,NO,FULL SCREEN=NO,AUTO-CONTINUE=NO, 

PNT1用の理論値を変更しますか? 

IF/C1.INPUT=="YES" 

CS1 =CALLSUB/CHANGETHEO,:, 

END_IF/ 

COMMENT/OPER,NO,FULL SCREEN=NO,AUTO-CONTINUE=NO, 

PNT1用のXYZ理論値、及び、実績値は: 

"Theo X= " + PNT1.TX 

"Theo Y= " + PNT1.TY 

"Theo Z= " + PNT1.TZ 

-------------------- 

"Actl X= " + PNT1.X 

"Actl Y= " + PNT1.Y 

"Actl Z= " + PNT1.Z 

ROUTINE/END 

SUBROUTINE/CHANGETHEO, 

POINT1 = {PNT1} : , 

= 

DIMINFO/PNT1;ICON,DIMID,FEATID,VERT,HORIZ, ,$ 

HEADINGS, ;MEAS, , , , , , , , 

C2 =COMMENT/INPUT,NO,FULL SCREEN=NO, 

PNT1用の新しいX理論値を入力して下さい。 

"現在の値は " + PNT1.TX 

ASSIGN/PNT1.TX=C2.INPUT 

C3 =COMMENT/INPUT,NO,FULL SCREEN=NO, 

PNT1用の新しいY理論値を入力して下さい。 

"現在の値は " + PNT1.TY 

ASSIGN/PNT1.TY=C3.INPUT 

C4 =COMMENT/INPUT,NO,FULL SCREEN=NO, 

PNT1用の新しいZ理論値を入力して下さい。 

"現在の値は " + PNT1.TZ 

ASSIGN/PNT1.TZ=C4.INPUT 

ENDSUB/ 

サンプル コードの説明 

C1=COMMENT/YESNO 
この行は、ユーザーからの、「はい」、または、「いいえ」の応答を受けて格納します。 
IF/C1.INPUT=="YES" 
この行は式です。これは、コメント1の入力が、「はい」であるかどうかを検査します。「はい」

である場合、IFステートメントは「真」であり、IFステートメント後のステートメントを実行し続

け、この場合、PNT1 

要素が測定されます。「いいえ」の場合、END_IFステートメントに移動します。 
CS1=CALLSUB/CHANGETHEO,:, 
この行では、CHANGETHEOという名のサブルーチンを呼び出します。ここで測定ルーチンのフロ

ーは、SUBROUTINE/CHANGETHEO行にジャンプします。 
SUBROUTINE/CHANGETHEO 
この行は、CHANGETHEOサブルーチンを初期化します。測定ルーチンのフローは、この行とEND

SUB/ 行の間のコードを実行し続けます。 
POINT1 = {PNT1} : , 
これは、サブルーチンの唯一の引数です。これによって、サブルーチンは、PNT1要素の情報にア

クセスできます。 
C2=COMMENT/INPUT, C3=COMMENT/INPUT, C4=COMMENT/INPUT 
これらの入力コメントはすべて、ユーザーからのX、Y、及び、Zの新しい理論値を受け取り、それ

らをそれぞれC2.INPUT、C3.INPUT、及び、C4.INPUTに保存します。 
ASSIGN/PNT1.TX = C2.INPUT 
このラインは、C2.INPUTからXの理論値を取り、それをPNT1.TX変数に割り当てます。PNT1.TX

はPC-DMISの変数で、PNT1のIDラベルと共に、Xの理論値（TXで示される）を持ちます。 
ASSIGN/PNT1.TY = C3.INPUT 
この行は、C3.INPUTからYの理論値を取り、それをPNT1.TY変数に割り当てます。PNT1.TYはPC
-DMISの変数で、PNT1のIDラベルと共に、Yの理論値（TYで示される）を持ちます。 
ASSIGN/PNT1.TZ = C4.INPUT 
この行は、C4.INPUTからZの理論値を取り、それをPNT1.TZ変数に割り当てます。PNT1.TZはPC
-DMISの変数で、PNT1のIDラベルと共に、Zの理論値（TZで示される）を持ちます。 
ENDSUB/ 
この行はサブルーチンを終了し、測定ルーチン 
フローは、サブルーチンの呼び出し直後のラインに戻ります。この場合、それはEND_IF/ステー

トメントです。その後、測定ルーチン 
フローは、X、Y、及び、Zの理論値と実績値を表示する次のオペレーター 
コメントを実行し続け、それから、このパーツプログラムは、ROUTINE/ENDコマンドによって終

了します。 

測定ルーチンの終了 

挿入 | フロー制御コマンド| ルーチンの終了メニュー 
オプションを選ぶと、ROUTINE/ENDコマンドが、編集ウィンドウ内に挿入されます。PC-
DMISが測定ルーチンの実行中にこのコマンドに遭遇するとき、いつでも直ちに測定ルーチンの実

行を中止します。 

あらかじめ定義された条件に基づいて、通常より早く測定ルーチンを終えたい時、このコマンドは

役に立ちます。 

